Доброго времени суток, есть запрос выводящий информацию из справочника, код запроса прилагаю ниже.

Текст запроса
```1c
    ВЫБРАТЬ
    ПриборыУчета.Наименование КАК Наименование,
    ПриборыУчета.Услуга КАК Услуга,
    ПриборыУчета.ЗаводскойНомер КАК ЗаводскойНомер,
    ПриборыУчета.ЕдиницаИзмерения КАК ЕдиницаИзмерения,
    КОЛИЧЕСТВО(ПриборыУчета.Ссылка) КАК КоличествоПовторений
    ИЗ
    Справочник.ПриборыУчета КАК ПриборыУчета

    СГРУППИРОВАТЬ ПО
    ПриборыУчета.ЗаводскойНомер,
    ПриборыУчета.Наименование,
    ПриборыУчета.Услуга,
    ПриборыУчета.ЕдиницаИзмерения

    ИМЕЮЩИЕ
    КОЛИЧЕСТВО(ПриборыУчета.Ссылка) > 1

    УПОРЯДОЧИТЬ ПО
    КоличествоПовторений УБЫВ
```

Так случилось, что требуется ещё ссылка на прибор учета.
Но при добавлении
```1c
    ПриборыУчета.СсылкаКАК Ссылка,
```
 запрос перестает работать во все, а именно выводит ничего.
Мне кажется это из-за использования функции КОЛИЧЕСТВО.
Сам запрос выбирает элементы у которых повторяются записи с Заводским Номером и выводит доп. колонку о кол-ве повторений. Прилагаю скриншот.
Есть вариант решения?
---
Есть изъян в постановке задачи. С одной стороны Вы хотите получить количество повторений некоторого набора значений реквизитов справочника. С другой стороны - получить ссылку на прибор учета. Так на какой прибор из нескольких дублей Вам нужна ссылка? Решение будет зависеть от ответа на поставленный вопрос.

Если нужна любая из ссылок, то задача тривиальна:

```1c
    ВЫБРАТЬ
    ПриборыУчета.Наименование КАК Наименование,
    ПриборыУчета.Услуга КАК Услуга,
    ПриборыУчета.ЗаводскойНомер КАК ЗаводскойНомер,
    ПриборыУчета.ЕдиницаИзмерения КАК ЕдиницаИзмерения,
    КОЛИЧЕСТВО(ПриборыУчета.Ссылка) КАК КоличествоПовторений,
    МАКСИМУМ(ПриборыУчета.Ссылка) КАК ЛюбаяИзСсылок
    ИЗ
    Справочник.ПриборыУчета КАК ПриборыУчета

    СГРУППИРОВАТЬ ПО
    ПриборыУчета.ЗаводскойНомер,
    ПриборыУчета.Наименование,
    ПриборыУчета.Услуга,
    ПриборыУчета.ЕдиницаИзмерения

    ИМЕЮЩИЕ
    КОЛИЧЕСТВО(ПриборыУчета.Ссылка) > 1

    УПОРЯДОЧИТЬ ПО
    КоличествоПовторений УБЫВ
```

Если нужны все ссылки, то можно представить список в виде дерева:

```1c
    ВЫБРАТЬ
    ПриборыУчета.Наименование КАК Наименование,
    ПриборыУчета.Услуга КАК Услуга,
    ПриборыУчета.ЗаводскойНомер КАК ЗаводскойНомер,
    ПриборыУчета.ЕдиницаИзмерения КАК ЕдиницаИзмерения
    ПОМЕСТИТЬ ВтДубли
    ИЗ
    Справочник.ПриборыУчета КАК ПриборыУчета

    СГРУППИРОВАТЬ ПО
    ПриборыУчета.ЗаводскойНомер,
    ПриборыУчета.Наименование,
    ПриборыУчета.Услуга,
    ПриборыУчета.ЕдиницаИзмерения

    ИМЕЮЩИЕ
    КОЛИЧЕСТВО(ПриборыУчета.Ссылка) > 1
    ;

    ВЫБРАТЬ
    ПриборыУчета.Наименование КАК Наименование,
    ПриборыУчета.Услуга КАК Услуга,
    ПриборыУчета.ЗаводскойНомер КАК ЗаводскойНомер,
    ПриборыУчета.ЕдиницаИзмерения КАК ЕдиницаИзмерения,
    ПриборыУчета.Ссылка КАК Ссылка,
    ИЗ
    Справочник.ПриборыУчета КАК ПриборыУчета
    ВНУТРЕННЕЕ СОЕДИНЕНИЕ ВтДубли КАК ВтДубли
    ПО ПриборыУчета.Наименование = ВтДубли.Наименование
    И ПриборыУчета.Услуга = ВтДубли.Услуга
    И ПриборыУчета.ЗаводскойНомер = ВтДубли.ЗаводскойНомер
    И ПриборыУчета.ЕдиницаИзмерения = ВтДубли.ЕдиницаИзмерения

    ИТОГИ
    КОЛИЧЕСТВО(Ссылка)
    ПО
    Наименование,
    Услуга,
    ЗаводскойНомер,
    ЕдиницаИзмерения
```
---
ПриборыУчета.СсылкаКАК Ссылка,
Это просто опечатка, или так и в запросе?
Помимо добавления поля, его еще в группировку добавить нужно.

И да, в таком случае будет КОЛИЧЕСТВО(ПриборыУчета.Ссылка) всегда = 1.
---
Опечатка.
Очевидно, что нужно добавлять, иначе ругаться будет. Добавлено, один чёрт не выводит.

Почему равно 1? мне нужно вывести только те где больше 1.
---
Почему равно 1?
Потому Количество(Ссылки) в Группировке по Ссылке (в данном случае) всегда будет 1.

мне нужно вывести только те где больше 1
см. (3)
---
удалено, не работающее.

Только учтите, что количество выводимых строк увеличится.
И для каждой выводимой ссылки будет указываться КоличествоПовторений не ссылок.
---
Есть изъян в постановке задачи. С одной стороны Вы хотите получить количество повторений некоторого набора значений реквизитов справочника. С другой стороны - получить ссылку на прибор учета. Так на какой прибор из нескольких дублей Вам нужна ссылка? Решение будет зависеть от ответа на поставленный вопрос.

Если нужна любая из ссылок, то задача тривиальна:

```1c
    ВЫБРАТЬ
    ПриборыУчета.Наименование КАК Наименование,
    ПриборыУчета.Услуга КАК Услуга,
    ПриборыУчета.ЗаводскойНомер КАК ЗаводскойНомер,
    ПриборыУчета.ЕдиницаИзмерения КАК ЕдиницаИзмерения,
    КОЛИЧЕСТВО(ПриборыУчета.Ссылка) КАК КоличествоПовторений,
    МАКСИМУМ(ПриборыУчета.Ссылка) КАК ЛюбаяИзСсылок
    ИЗ
    Справочник.ПриборыУчета КАК ПриборыУчета

    СГРУППИРОВАТЬ ПО
    ПриборыУчета.ЗаводскойНомер,
    ПриборыУчета.Наименование,
    ПриборыУчета.Услуга,
    ПриборыУчета.ЕдиницаИзмерения

    ИМЕЮЩИЕ
    КОЛИЧЕСТВО(ПриборыУчета.Ссылка) > 1

    УПОРЯДОЧИТЬ ПО
    КоличествоПовторений УБЫВ
```

Если нужны все ссылки, то можно представить список в виде дерева:

```1c
    ВЫБРАТЬ
    ПриборыУчета.Наименование КАК Наименование,
    ПриборыУчета.Услуга КАК Услуга,
    ПриборыУчета.ЗаводскойНомер КАК ЗаводскойНомер,
    ПриборыУчета.ЕдиницаИзмерения КАК ЕдиницаИзмерения
    ПОМЕСТИТЬ ВтДубли
    ИЗ
    Справочник.ПриборыУчета КАК ПриборыУчета

    СГРУППИРОВАТЬ ПО
    ПриборыУчета.ЗаводскойНомер,
    ПриборыУчета.Наименование,
    ПриборыУчета.Услуга,
    ПриборыУчета.ЕдиницаИзмерения

    ИМЕЮЩИЕ
    КОЛИЧЕСТВО(ПриборыУчета.Ссылка) > 1
    ;

    ВЫБРАТЬ
    ПриборыУчета.Наименование КАК Наименование,
    ПриборыУчета.Услуга КАК Услуга,
    ПриборыУчета.ЗаводскойНомер КАК ЗаводскойНомер,
    ПриборыУчета.ЕдиницаИзмерения КАК ЕдиницаИзмерения,
    ПриборыУчета.Ссылка КАК Ссылка,
    ИЗ
    Справочник.ПриборыУчета КАК ПриборыУчета
    ВНУТРЕННЕЕ СОЕДИНЕНИЕ ВтДубли КАК ВтДубли
    ПО ПриборыУчета.Наименование = ВтДубли.Наименование
    И ПриборыУчета.Услуга = ВтДубли.Услуга
    И ПриборыУчета.ЗаводскойНомер = ВтДубли.ЗаводскойНомер
    И ПриборыУчета.ЕдиницаИзмерения = ВтДубли.ЕдиницаИзмерения

    ИТОГИ
    КОЛИЧЕСТВО(Ссылка)
    ПО
    Наименование,
    Услуга,
    ЗаводскойНомер,
    ЕдиницаИзмерения
```
---
```1c
    ВЫБРАТЬ
    ПриборыУчета.Ссылка КАК Ссылка,
    ПриборыУчета.Наименование КАК Наименование,
    ПриборыУчета.Услуга КАК Услуга,
    ПриборыУчета.ЗаводскойНомер КАК ЗаводскойНомер,
    ПриборыУчета.ЕдиницаИзмерения КАК ЕдиницаИзмерения,
    КОЛИЧЕСТВО(ПриборыУчетаПовторения.Ссылка) КАК КоличествоПовторений
    ИЗ
    Справочник.ПриборыУчета КАК ПриборыУчета
    ЛЕВОЕ СОЕДИНЕНИЕ Справочник.ПриборыУчета КАК ПриборыУчетаПовторения
    ПО ПриборыУчета.Наименование = ПриборыУчетаПовторения.Наименование
    И ПриборыУчета.Услуга = ПриборыУчетаПовторения.Услуга
    И ПриборыУчета.ЗаводскойНомер = ПриборыУчетаПовторения.ЗаводскойНомер
    И ПриборыУчета.ЕдиницаИзмерения = ПриборыУчетаПовторения.ЕдиницаИзмерения

    СГРУППИРОВАТЬ ПО
    ПриборыУчета.Ссылка,
    ПриборыУчета.ЗаводскойНомер,
    ПриборыУчета.Наименование,
    ПриборыУчета.Услуга,
    ПриборыУчета.ЕдиницаИзмерения

    ИМЕЮЩИЕ
    КОЛИЧЕСТВО(ПриборыУчетаПовторения.Ссылка) > 1

    УПОРЯДОЧИТЬ ПО
    КоличествоПовторений УБЫВ
```
---
Попробуйте так:
Максимум(ПриборыУчета.Ссылка)
и
Убрать группировку по ссылке